

NAG Library Routine Document

E04FYF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

E04FYF is an easy-to-use algorithm for finding an unconstrained minimum of a sum of squares of m nonlinear functions in n variables ($m \geq n$). No derivatives are required.

It is intended for functions which are continuous and which have continuous first and second derivatives (although it will usually work even if the derivatives have occasional discontinuities).

2 Specification

```
SUBROUTINE E04FYF(M, N, LSFUN1, X, FSUMSQ, W, LW, IUSER, RUSER, IFAIL)
INTEGER          M, N, LW, IUSER(*), IFAIL
double precision X(N), FSUMSQ, W(LW), RUSER(*)
EXTERNAL         LSFUN1
```

3 Description

E04FYF is essentially identical to the subroutine LSNDN1 in the NPL Algorithms Library. It is applicable to problems of the form

$$\text{Minimize } F(x) = \sum_{i=1}^m [f_i(x)]^2$$

where $x = (x_1, x_2, \dots, x_n)^T$ and $m \geq n$. (The functions $f_i(x)$ are often referred to as 'residuals'.)

You must supply a subroutine to evaluate functions $f_i(x)$ at any point x .

From a starting point supplied by you, a sequence of points is generated which is intended to converge to a local minimum of the sum of squares. These points are generated using estimates of the curvature of $F(x)$.

4 References

Gill P E and Murray W (1978) Algorithms for the solution of the nonlinear least-squares problem *SIAM J. Numer. Anal.* **15** 977–992

5 Parameters

1: M – INTEGER *Input*
 2: N – INTEGER *Input*

On entry: the number m of residuals, $f_i(x)$, and the number n of variables, x_j .

Constraint: $1 \leq N \leq M$.

3: LSFUN1 – SUBROUTINE, supplied by the user. *External Procedure*

You must supply this routine to calculate the vector of values $f_i(x)$ at any point x . It should be tested separately before being used in conjunction with E04FYF (see the E04 Chapter Introduction).

The specification of LSFUN1 is:

```
SUBROUTINE LSFUN1(M, N, XC, FVEC, IUSER, RUSER)
```

```
INTEGER          M, N, IUSER(*)
```

```
double precision XC(N), FVEC(M), RUSER(*)
```

1: M – INTEGER

Input

2: N – INTEGER

Input

On entry: the numbers m and n of residuals and variables, respectively.

3: XC(N) – **double precision** array

Input

On entry: the point x at which the values of the f_i are required.

4: FVEC(M) – **double precision** array

Output

On exit: FVEC(i) must contain the value of f_i at the point x , for $i = 1, 2, \dots, m$.

5: IUSER(*) – INTEGER array

User Workspace

6: RUSER(*) – **double precision** array

User Workspace

LSFUN1 is called from E04FYF with the parameters IUSER and RUSER as supplied to E04FYF. You are free to use the arrays IUSER and RUSER to supply information to LSFUN1 as an alternative to using COMMON.

LSFUN1 must be declared as EXTERNAL in the (sub)program from which E04FYF is called. Parameters denoted as *Input* must **not** be changed by this procedure.

4: X(N) – **double precision** array

Input/Output

On entry: X(j) must be set to a guess at the j th component of the position of the minimum, for $j = 1, 2, \dots, n$.

On exit: the lowest point found during the calculations. Thus, if IFAIL = 0 on exit, X(j) is the j th component of the position of the minimum.

5: FSUMSQ – **double precision**

Output

On exit: the value of the sum of squares, $F(x)$, corresponding to the final point stored in X.

6: W(LW) – **double precision** array

Workspace

7: LW – INTEGER

Input

On entry: the dimension of the array W as declared in the (sub)program from which E04FYF is called.

Constraints:

if $N > 1$, $LW \geq 7 \times N + N \times N + 2 \times M \times N + 3 \times M + N \times (N - 1)/2$;

if $N = 1$, $LW \geq 9 + 5 \times M$.

8: IUSER(*) – INTEGER array

User Workspace

Note: the dimension of the array IUSER must be at least 1.

IUSER is not used by E04FYF, but is passed directly to LSFUN1 and may be used to pass information to that routine.

9: RUSER(*) – **double precision** array

User Workspace

Note: the dimension of the array RUSER must be at least 1.

RUSER is not used by E04FYF, but is passed directly to LSFUN1 and may be used to pass information to that routine.

10: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1 . If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq 0$ on exit, the recommended value is -1 . **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1 , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Note: E04FYF may return useful information for one or more of the following detected errors or warnings.

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $N < 1$,
 or $M < N$,
 or $LW < 7 \times N + N \times N + 2 \times M \times N + 3 \times M + N \times (N - 1)/2$, when $N > 1$,
 or $LW < 9 + 5 \times M$, when $N = 1$.

IFAIL = 2

There have been $400 \times n$ calls of LSFUN1, yet the algorithm does not seem to have converged. This may be due to an awkward function or to a poor starting point, so it is worth restarting E04FYF from the final point held in X.

IFAIL = 3

The final point does not satisfy the conditions for acceptance as a minimum, but no lower point could be found.

IFAIL = 4

An auxiliary routine has been unable to complete a singular value decomposition in a reasonable number of sub-iterations.

IFAIL = 5

IFAIL = 6

IFAIL = 7

IFAIL = 8

There is some doubt about whether the point x found by E04FYF is a minimum of $F(x)$. The degree of confidence in the result decreases as IFAIL increases. Thus, when IFAIL = 5, it is probable that the final x gives a good estimate of the position of a minimum but, when IFAIL = 8, it is very unlikely that the routine has found a minimum.

If you are not satisfied with the result (e.g., because IFAIL lies between 3 and 8), it is worth restarting the calculations from a different starting point (not the point at which the failure occurred) in order to avoid the region which caused the failure. Repeated failure may indicate some defect in the formulation of the problem.

7 Accuracy

If the problem is reasonably well scaled and a successful exit is made, then, for a computer with a mantissa of t decimals, one would expect to get about $t/2 - 1$ decimals accuracy in the components of x and between $t - 1$ (if $F(x)$ is of order 1 at the minimum) and $2t - 2$ (if $F(x)$ is close to zero at the minimum) decimals accuracy in $F(x)$.

8 Further Comments

The number of iterations required depends on the number of variables, the number of residuals and their behaviour, and the distance of the starting point from the solution. The number of multiplications performed per iteration of E04FYF varies, but for $m \gg n$ is approximately $n \times m^2 + O(n^3)$. In addition, each iteration makes at least $n + 1$ calls of LSFUN1. So, unless the residuals can be evaluated very quickly, the run time will be dominated by the time spent in LSFUN1.

Ideally, the problem should be scaled so that the minimum value of the sum of squares is in the range $(0, +1)$, and so that at points a unit distance away from the solution the sum of squares is approximately a unit value greater than at the minimum. It is unlikely that you will be able to follow these recommendations very closely, but it is worth trying (by guesswork), as sensible scaling will reduce the difficulty of the minimization problem, so that E04FYF will take less computer time.

When the sum of squares represents the goodness-of-fit of a nonlinear model to observed data, elements of the variance-covariance matrix of the estimated regression coefficients can be computed by a subsequent call to E04YCF, using information returned in segments of the workspace array W. See E04YCF for further details.

9 Example

This example finds least-squares estimates of x_1 , x_2 and x_3 in the model

$$y = x_1 + \frac{t_1}{x_2 t_2 + x_3 t_3}$$

using the 15 sets of data given in the following table.

y	t_1	t_2	t_3
0.14	1.0	15.0	1.0
0.18	2.0	14.0	2.0
0.22	3.0	13.0	3.0
0.25	4.0	12.0	4.0
0.29	5.0	11.0	5.0
0.32	6.0	10.0	6.0
0.35	7.0	9.0	7.0
0.39	8.0	8.0	8.0
0.37	9.0	7.0	7.0
0.58	10.0	6.0	6.0
0.73	11.0	5.0	5.0
0.96	12.0	4.0	4.0
1.34	13.0	3.0	3.0
2.10	14.0	2.0	2.0
4.39	15.0	1.0	1.0

The program uses $(0.5, 1.0, 1.5)$ as the initial guess at the position of the minimum.

9.1 Program Text

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

```

*      E04FYF Example Program Text
*      Mark 19 Revised. NAG Copyright 1999.
*      .. Parameters ..
      INTEGER          N, M, NT, LW
      PARAMETER        (N=3,M=15,NT=3,LW=7*N+N*N+2*M*N+3*M+N*(N-1)/2)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      DOUBLE PRECISION FSUMSQ
      INTEGER          I, IFAIL, J, K
*      .. Local Arrays ..
      DOUBLE PRECISION RUSER(M+M*NT), T(M,NT), W(LW), X(N), Y(M)
      INTEGER          IUSER(1)
*      .. External Subroutines ..
      EXTERNAL         E04FYF, LSFUN1
*      .. Executable Statements ..
      WRITE (NOUT,*) 'E04FYF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)

*
*      Observations of TJ (J = 1, 2, 3) are held in T(I, J)
*      (I = 1, 2, . . . , 15)
*
      IUSER(1) = NT
      K = M
      DO 40 I = 1, M
        READ (NIN,*) Y(I), (T(I,J),J=1,NT)
        RUSER(I) = Y(I)
        DO 20 J = 1, NT
          RUSER(K+J) = T(I,J)
20      CONTINUE
        K = K + NT
40     CONTINUE
*
      X(1) = 0.5D0
      X(2) = 1.0D0
      X(3) = 1.5D0
*
      IFAIL = 1
*
      CALL E04FYF(M,N,LSFUN1,X,FSUMSQ,W,LW,IUSER,RUSER,IFAIL)
*
      IF (IFAIL.LT.0) THEN
        WRITE (NOUT,*)
        WRITE (NOUT,99997) ' ** E04FYF returned with IFAIL = ', IFAIL
      ELSE
        IF (IFAIL.NE.0) THEN
          WRITE (NOUT,*)
          WRITE (NOUT,99999) 'Error exit type', IFAIL,
+          ' - see routine document'
        END IF
*
        IF (IFAIL.NE.1) THEN
          WRITE (NOUT,*)
          WRITE (NOUT,99998) 'On exit, the sum of squares is', FSUMSQ
          WRITE (NOUT,99998) 'at the point', (X(J),J=1,N)
        END IF
      END IF
*
99999 FORMAT (1X,A,I3,A)
99998 FORMAT (1X,A,3F12.4)
99997 FORMAT (1X,A,I5)
      END
*
      SUBROUTINE LSFUN1(M,N,XC,FVEC,IUSER,RUSER)
*      .. Scalar Arguments ..

```

```

      INTEGER          M, N
*      .. Array Arguments ..
      DOUBLE PRECISION FVEC(M), RUSER(*), XC(N)
      INTEGER          IUSER(*)
*      .. Local Scalars ..
      INTEGER          I, K
*      .. Executable Statements ..
      K = M
      DO 20 I = 1, M
         FVEC(I) = XC(1) + RUSER(K+1)/(XC(2)*RUSER(K+2)+XC(3)*RUSER(K+3)
+           ) - RUSER(I)
         K = K + IUSER(1)
      20 CONTINUE
      RETURN
      END

```

9.2 Program Data

E04FYF Example Program Data

0.14	1.0	15.0	1.0
0.18	2.0	14.0	2.0
0.22	3.0	13.0	3.0
0.25	4.0	12.0	4.0
0.29	5.0	11.0	5.0
0.32	6.0	10.0	6.0
0.35	7.0	9.0	7.0
0.39	8.0	8.0	8.0
0.37	9.0	7.0	7.0
0.58	10.0	6.0	6.0
0.73	11.0	5.0	5.0
0.96	12.0	4.0	4.0
1.34	13.0	3.0	3.0
2.10	14.0	2.0	2.0
4.39	15.0	1.0	1.0

9.3 Program Results

E04FYF Example Program Results

On exit, the sum of squares is 0.0082
 at the point 0.0824 1.1330 2.3437
